Series-Shunt Amplifier Bias Analysis
Worst Case Analysis Illustration

There are several useful statistical techniques available in PSpice (in particular); Monte Carlo and Worst Case analyses are two principal cases. In both cases some or all circuit parameters are assigned tolerances. In a representative Monte Carlo analysis a circuit is analysed repeatedly (for a specified number of times); for each analysis parameter values are chosen at random within the specified part tolerance. When the computations are completed the high, low, and average values of parameters are reported. Monte Carlo analysis simulates sampling on a production line to evaluate the range of performance within which a product will function. Thus, for example, if 80% of a production run will function satisfactorily it may be advantageous to test each product and discard those with substandard performance rather than, say, tighten tolerances at added costs.

On the other hand circumstances may dictate 100% satisfactory performance. For example satellite electronics are difficult and costly to repair after launch. In this case a ‘worst case’ analysis provides some assurance against premature degradation of performance. The computation is done a bit differently in this case. The sensitivity of the circuit performance to variation in each part with a tolerance is computed, and this information is used to choose parameter values within a tolerance to lead to extreme performance (max. and min.).

This note presents a worst case analysis of the biasing stability of the shunt-series feedback pair illustrated to the right; for a transistor \( \beta \) tolerance of \( \pm 40\% \) and resistor tolerance of \( \pm 10\% \).

The analysis command is:
```
.WCASE  <ANALYSIS TYPE>  <OUTPUT VARIABLE>  <FUNCTION>
```

The specific command form used below is:
```
.WCASE DC V(4) YMAX ;YMAX calls the greatest difference from nominal value
```

In addition to the analysis command tolerances must be specified for those parts that are to have them. The .MODEL command is used for this, e.g.,
```
.MODEL Q2N3904  NPN (BF=460 DEV 40\%)
.MODEL RMOD     RES  (R=1 DEV 10\%)
```

Nominal Node Voltages
(1) 10.0000 (2) 3.0680 (3) 7744 (4) 5.8131 (5) .2.2877

WORST CASE ALL DEVICES‡
These data are for the DC analysis

<table>
<thead>
<tr>
<th>DEVICE</th>
<th>MODEL</th>
<th>PARAMETER</th>
<th>NEW VALUE</th>
</tr>
</thead>
<tbody>
<tr>
<td>Q1</td>
<td>Q2N3904</td>
<td>BF</td>
<td>168 (Increased)</td>
</tr>
<tr>
<td>Q2</td>
<td>Q2N3904</td>
<td>BF</td>
<td>72 (Decreased)</td>
</tr>
<tr>
<td>RC1</td>
<td>RMOD</td>
<td>R</td>
<td>1.1 (Increased)</td>
</tr>
<tr>
<td>RC2</td>
<td>RMOD</td>
<td>R</td>
<td>.9 (Decreased)</td>
</tr>
<tr>
<td>RE2</td>
<td>RMOD</td>
<td>R</td>
<td>1.1 (Increased)</td>
</tr>
<tr>
<td>RFB</td>
<td>RMOD</td>
<td>R</td>
<td>.9 (Decreased)</td>
</tr>
</tbody>
</table>
‡ R parameter is factor by which nominal resistance is multiplied. BF (DC beta) is actual
BJT model parameter value (nominal = 120)

WORST CASE ALL DEVICES
V(4) = 7.42 volts

MAX DEVIATION FROM NOMINAL ALL DEVICES
1.6113 volts higher at VCC = 10 (127.72% of Nominal)

*Series-Shunt Amplifier
VCC 1 0 DC 10V
RC1 1 2 RMOD 6.8K
Q1 2 3 0 Q2N3904
RC2 1 4 RMOD 3.3K
Q2 4 2 5 Q2N3904
RE2 5 0 RMOD 1.8K
RFB 5 3 RMOD 180K
.LIB EVAL.LIB
.OP
.PROBE
.WCASE DC V(4) YMAX
.MODEL Q2N3904 NPN (BF=460 DEV 40%)
.MODEL RMOD RES (R=1 DEV 10%)

* Only one.WCASE analysis allowed at one time. Note: for simplicity the DC analysis
* increment is larger then the start-end interval, so that only the starting value VCC=10V will be
* used.
.DC VCC 10 10.5 1
.END